Solving equations

Basic solving

Use the **solve** command to solve an equation or system of equations.

```
> solve(x^2-5*x+6=0,x);
```

$$3, 2 \tag{1.1}$$

It is usually best to put the variable in curly brackets. This means that Maple will include the variable name when reporting the solution.

```
> solve(x^2-5*x+6=0,{x});
```

$$\{x=3\}, \{x=2\} \tag{1.2}$$

When there is only one variable involved, it is not strictly necessary to specify it. However, if two variables are involved, then Maple may guess wrongly about which one to solve for. For example, if we ask Maple to solve $a\,x^2+b\,x+c=0$, it will give the formula $c=-a\,x^2-b\,x$ (expressing c in terms of a, b and x) rather than the usual quadratic formula expressing x in terms of a, b and c.

```
> solve(a*x^2+b*x+c=0);
```

$$\{a=a,\, b=b,\, c=-a\,x^2-b\,x,\, x=x\} \tag{1.3}$$

```
> solve(a*x^2+b*x+c=0,{x});
```

$$\left\{x=\frac{-b+\sqrt{-4\,a\,c+b^2}}{2\,a}\right\}, \left\{x=-\frac{b+\sqrt{-4\,a\,c+b^2}}{2\,a}\right\} \tag{1.4}$$

Here is an example with two equations and two variables:

```
> solve({x+y=4*a,x-y=4*b},{x,y});
```

$$\{x=2\,b+2\,a,\, y=-2\,b+2\,a\} \tag{1.5}$$

Now suppose we want to substitute these values into the expression x^3-y^3. For this, we use the **subs** function.

```
> subs({y = 2*a-2*b, x = 2*a+2*b},x^3-y^3);
```

$$(2\,b+2\,a)^3-(-2\,b+2\,a)^3 \tag{1.6}$$

```
> expand(%);
```

$$48\,a^2\,b+16\,b^3 \tag{1.7}$$

Here is a slightly better way:

```
> sol := solve({x+y=4*a,x-y=4*b},{x,y});
```

$$sol := \{x=2\,b+2\,a,\, y=-2\,b+2\,a\} \tag{1.8}$$

```
> expand(subs(sol,x^3-y^3));
```

$$48\,a^2\,b+16\,b^3 \tag{1.9}$$

Solve the equation

$$x^2 = \frac{1}{2+x^2}$$

We first plot the graphs of x^2 and $\frac{1}{2+x^2}$, and observe that there are two crossing points:

```
> plot([x^2,1/(2+x^2)],x=-2..2);
```



Alternatively, we can plot $y = x^2 - \frac{1}{2+x^2}$, and check where the graph crosses the x-axis.

```
> plot(x^2-1/(2+x^2),x=-2..2);
```



We now ask Maple to solve the equation. It reports four different solutions, but only two can be seen from the graph.

```
> solve(x^2=1/(2+x^2),x);
```

$$\sqrt{\sqrt{2}-1},\ -\sqrt{\sqrt{2}-1},\ \mathrm{I}\sqrt{1+\sqrt{2}},\ -\mathrm{I}\sqrt{1+\sqrt{2}} \qquad \textbf{(2.1)}$$

If we evaluate numerically, we find that the two extra solutions are complex numbers.

```
> evalf(%);
```

$$0.6435942526,\ -0.6435942526,\ 1.553773974\,\mathrm{I},\ -1.553773974\,\mathrm{I} \qquad \textbf{(2.2)}$$

We can solve the equation by hand as follows. We start with the equation

$$x^2 = \frac{1}{2+x^2},$$

and multiply both sides by $2+x^2$ to get

$$x^4 + 2\,x^2 = 1.$$

We then note that

$$\left(x^2+1\right)^2 = x^4 + 2\,x^2 + 1 = 1 + 1 = 2.$$

As x^2+1 is certainly positive, there is no sign ambiguity, so we conclude that $x^2+1=\sqrt{2}$. This can be rearranged to give

$$x = \sqrt{\sqrt{2}-1} \quad \text{or} \quad x = -\sqrt{\sqrt{2}-1}.$$

Solve (separately) the equations $y=1$, $y=-1$, $y=0$ and $y=\cos(\pi x)$,

where $y = 1 - \frac{8x^2}{3} + \frac{2x^4}{3}$.

```
> y := 1-8*x^2/3+2*x^4/3;
```

$$y := 1 - \frac{8}{3}x^2 + \frac{2}{3}x^4 \quad (3.1)$$

```
> plot([y,1,-1,cos(Pi*x)],
       x=-3..3,-2..2,
       color=[red,green,green,blue]
  );
```



You can see from the picture that the red graph crosses the line $y = -1$ in four places. Maple finds these without any trouble.

```
> solve(y=-1,x);
```

$$1, -1, \sqrt{3}, -\sqrt{3} \quad (3.2)$$

You can also see that the red curve crosses the line $y = 1$ at $x = -2$ and $x = 2$, and that it just touches the line at $x = 0$. This means that $x = 0$ is a "double solution", so Maple reports it twice:

```
> solve(y=1,x);
```

$$0, 0, 2, -2 \quad (3.3)$$

The graph crosses the x-axis in four places. Although the formulae are more complicated, Maple again finds them without any trouble.

```
> solve(y=0,x);
```

$$\frac{\sqrt{8-2\sqrt{10}}}{2},\ -\frac{\sqrt{8-2\sqrt{10}}}{2},\ \frac{\sqrt{8+2\sqrt{10}}}{2},\ -\frac{\sqrt{8+2\sqrt{10}}}{2} \tag{3.4}$$

We can use the **evalf** function to get numerical answers:

```
> evalf(%);
```

$$0.6471948470,\ -0.6471948470,\ 1.892389714,\ -1.892389714 \tag{3.5}$$

The solutions of $y = \cos(\text{Pi}\, x)$ are the five places where the red graph crosses the blue one, namely $x = -2$, $x = -1$, $x = 0$, $x = 1$ and $x = 2$. Maple is not clever enough to work this out symbolically.

```
> solve(y=cos(Pi*x),x);
```

$$\frac{\mathit{RootOf}\left(-3\cos(\_Z)\,\pi^4 + 3\,\pi^4 - 8\,\pi^2\,\_Z^2 + 2\,\_Z^4\right)}{\pi} \tag{3.6}$$

If Maple gives an answer involving **RootOf**, it often helps to set **_EnvExplicit := true;** In this case, however, this makes no difference:

```
> _EnvExplicit := true;
```

$$\_EnvExplicit := true \tag{3.7}$$

```
> solve(y=cos(Pi*x),x);
```

$$\frac{\mathit{RootOf}\left(-3\cos(\_Z)\,\pi^4 + 3\,\pi^4 - 8\,\pi^2\,\_Z^2 + 2\,\_Z^4\right)}{\pi} \tag{3.8}$$

However, we can see the solution graphically, either from our earlier picture, or more easily from the following one:

```
> plot(y-cos(Pi*x),x=-3..3,-2..5);
```




```
> unassign('y'):
```

Solve the equations $y = \cos\left(\frac{\pi x}{2}\right)$ and $x^2 + y^2 = 1$.

We start by plotting the curves $y = \cos\left(\frac{\pi x}{2}\right)$ and $x^2 + y^2 = 1$, to give an idea of what is going on.
The syntax for the first one is easy:

```
> plot(cos(Pi*x/2),x=-3..3);
```



The curve $x^2 + y^2 = 1$ is more difficult, because y is not given explicitly as a function of x. We therefore need the **implicitplot** function. This is not automatically available when Maple starts up; we have to load up the **plots** package first:

```
> with(plots);
```

$[animate, animate3d, animatecurve, arrow, changecoords, complexplot, complexplot3d, conformal, conformal3d, contourplot, contourplot3d, coordplot, coordplot3d, densityplot, display, dualaxisplot, fieldplot, fieldplot3d, gradplot, gradplot3d, implicitplot, implicitplot3d, inequal, interactive, interactiveparams, intersectplot, listcontplot, listcontplot3d, listdensityplot, listplot, listplot3d, loglogplot, logplot, matrixplot, multiple, odeplot, pareto, plotcompare, pointplot, pointplot3d, polarplot, polygonplot, polygonplot3d, polyhedra\_supported, polyhedraplot, rootlocus, semilogplot, setcolors, setoptions, setoptions3d, shadebetween, spacecurve, sparsematrixplot, surfdata, textplot, textplot3d, tubeplot]$ **(4.1)**

```
> implicitplot(
    x^2+y^2=1,
    x=-1..1,y=-1..1
  );
```



Our next problem is to draw the two graphs together. It is convenient to start by giving them names, and drawing them in different colours. Note that the commands below end with a colon, not a semicolon. This prevents Maple from printing out a big list of coordinates.

```
> p1 := plot(cos(Pi*x/2),
             x=-3..3,
             colour=red):
```

```
> p2 := implicitplot(x^2+y^2=1,
                     x=-1..1,y=-1..1,
                     colour=blue):
```

```
> display(p1,p2,scaling=constrained);
```



We see that there are three solutions: $[x, y] = [-1, 0]$, $[x, y] = [0, 1]$ and $[x, y] = [1, 0]$. Maple does not succeed in finding these symbolically.

```
> unassign('p1','p2'):
```

Find the points where the curve $[x, y] = \left[\frac{10\,t}{3\,(1+t^2)}, \frac{5\,(1-t^2)}{4\,(1+t^2)}\right]$ meets the curve $[x, y] = \left[\frac{5\,s}{2\,(1+s^2)}, \frac{5\,(1-s^2)}{3\,(1+s^2)}\right]$.

We start by giving names to the various formulae involved:

```
> x[1] := 10*t/(3*(1+t^2));
```

$$x_1 := \frac{10\,t}{3\,t^2+3} \tag{5.1}$$

```
> y[1] := 5*(1-t^2)/(4*(1+t^2));
```

$$y_1 := \frac{5\,(-t^2+1)}{4\,t^2+4} \tag{5.2}$$

```
> x[2] := 5*s/(2*(1+s^2));
```

$$x_2 := \frac{5\,s}{2\,s^2+2} \tag{5.3}$$

```
> y[2] := 5*(1-s^2)/(3*(1+s^2));
```

$$y_2 := \frac{5\,(-s^2+1)}{3\,s^2+3} \tag{5.4}$$

We next get an idea of what is going on, by plotting the two curves. We use the option **scaling=constrained** to make Maple use the same units on the two axes.

```
> plot([x[1],y[1],t=-10..10],scaling=constrained);
```




```
> plot([x[2],y[2],s=-10..10],scaling=constrained);
```



We really want to show both curves in the same picture. For this, we give names to the two pictures, then combine them. We also use different colours, so we can tell which curve is which.

Note that the following commands end with a colon, not a semicolon. This is to prevent Maple from printing out a huge list of coordinates.

```
> p[1] := plot([x[1],y[1],t=-10..10],
               scaling=constrained,
               colour=red):
```

```
> p[2] := plot([x[2],y[2],s=-10..10],
               scaling=constrained,
               colour=blue):
```

To combine the two plots, we need the **`display`** function from the **`plots`** package. The command **`with(plots)`** loads up the package. Maple responds by printing a list of all the new commands in the package; if you do not want to see the list, use a colon instead of a semicolon. The warning message is standard and can be ignored.

```
> with(plots);
```

[*animate*, *animate3d*, *animatecurve*, *arrow*, *changecoords*, *complexplot*, *complexplot3d*, *conformal*, *conformal3d*, *contourplot*, *contourplot3d*, *coordplot*, *coordplot3d*, *densityplot*, *display*, *dualaxisplot*, *fieldplot*, *fieldplot3d*, *gradplot*, *gradplot3d*, *implicitplot*, *implicitplot3d*, *inequal*, *interactive*, *interactiveparams*, *intersectplot*, *listcontplot*, *listcontplot3d*, *listdensityplot*, *listplot*, *listplot3d*, *loglogplot*, *logplot*, *matrixplot*, *multiple*, *odeplot*, *pareto*, *plotcompare*, *pointplot*, *pointplot3d*, *polarplot*, *polygonplot*, *polygonplot3d*, *polyhedra_supported*, *polyhedraplot*, *rootlocus*, **(5.5)** *semilogplot*, *setcolors*, *setoptions*, *setoptions3d*, *shadebetween*, *spacecurve*, *sparsematrixplot*, *surfdata*, *textplot*, *textplot3d*, *tubeplot*]

```
> display(p[1],p[2]);
```



There are evidently four points of intersection. To find their coordinates, we must solve the equations $x_1 = x_2$ and $y_1 = y_2$ for s and t. It will be convenient to give the name **sols** to the sequence of solutions.

```
> sols := solve({x[1]=x[2],y[1]=y[2]},{s,t});
```

$$sols := \{s=2, t=3\}, \left\{s=-\frac{1}{2}, t=-\frac{1}{3}\right\}, \left\{s=\frac{1}{2}, t=\frac{1}{3}\right\}, \{s=-2, t=-3\} \qquad \textbf{(5.6)}$$

We do not just want to know the values of s and t where the intersections occur, we want to know the x and y coordinates of the points of intersection. For example, the first intersection point occurs when $s=2$ and $t=3$, which means that

$$[x, y] = [x_1, y_1] = \left[\frac{10\,t}{3\,(1+t^2)}, \frac{5\,(1-t^2)}{4\,(1+t^2)}\right] = \left[\frac{30}{(3\cdot 10)}, -\frac{5\cdot 8}{(4\cdot 10)}\right] = [1, -1],$$

or alternatively

$$[x, y] = [x_2, y_2] = \left[\frac{5\,s}{2\,(1+s^2)}, \frac{5\,(1-s^2)}{3\,(1+s^2)}\right] = \left[\frac{10}{(2\cdot 5)}, -\frac{5\cdot 3}{(3\cdot 5)}\right] = [1, -1].$$

To get Maple to do these calculations, we enter

```
> subs(t=3,[x[1],y[1]]);
```

$$[1, -1] \tag{5.7}$$

```
> subs(s=2,[x[2],y[2]]);
```

$$[1, -1] \tag{5.8}$$

or

```
> subs({s=2,t=3},[x[1],y[1]]);
```

$$[1, -1] \tag{5.9}$$

or

```
> subs(sols[1],[x[1],y[1]]);
```

$$[1, -1] \tag{5.10}$$

```
> sols[1];
```

$$\{s=2, t=3\} \tag{5.11}$$

We find the other three intersection points in the same way:

```
> subs(sols[2],[x[1],y[1]]);
```

$$[-1, 1] \tag{5.12}$$

```
> subs(sols[3],[x[1],y[1]]);
```

$$[1, 1] \tag{5.13}$$

```
> subs(sols[4],[x[1],y[1]]);
```

$$[-1, -1] \tag{5.14}$$

To find all four intersection points at the same time, we use the **seq** command:

```
> seq(
   subs(sols[i],[x[1],y[1]]),
   i=1..4
  );
```

$$[1, -1], [-1, 1], [1, 1], [-1, -1] \tag{5.15}$$

```
> unassign('x','y','p'):
```

Solve the equation $\sin(x) = \sin\left(x + \frac{\pi}{3}\right)$

The obvious approach gives a solution:

```
> solve(sin(x)=sin(x+Pi/3),x);
```

$$\frac{\pi}{3} \tag{6.1}$$

However, this is not the only solution. To find the other ones, we set **_EnvAllSolutions:= true;**

```
> _EnvAllSolutions:=true;
```

$$\mathit{\_EnvAllSolutions} := \mathit{true} \tag{6.2}$$

```
> solve(sin(x)=sin(x+Pi/3),x);
```

$$\frac{1}{3}\pi+\pi\_Z1\sim \qquad \textbf{(6.3)}$$

In this answer, the symbol `_Z1~` stands for an arbitrary integer. Thus, the solutions are $x=\left(n+\frac{1}{3}\right)\mathrm{Pi}$ for integers n. More explicitly, the positive solutions are $x=\frac{4\,\mathrm{Pi}}{3}$, $x=\frac{7\,\mathrm{Pi}}{3}$, $x=\frac{10\,\mathrm{Pi}}{3}$ and so on. There are several ways to see this graphically. Firstly, we can plot the graphs $y=\sin(x)$ and $y=\sin\left(x+\frac{\mathrm{Pi}}{3}\right)$,

```
> plot([sin(x),sin(x+Pi/3)],x=-2*Pi..2*Pi);
```



This graph is not very illuminating, because the marks on the x-axis are at $x=-6, -5\,..4, 5, 6$ rather than at multiples of Pi. The easiest way to fix this is to put $x=\mathrm{Pi}\,t$ and plot everything in terms of t.

```
> x := Pi*t;
```

$$x:=\pi\,t \qquad \textbf{(6.4)}$$

```
> plot([sin(x),sin(x+Pi/3)],t=-2..2);
```



It is easier to see the solutions by plotting $\sin(x) - \sin\left(x + \frac{\text{Pi}}{3}\right)$ instead, and seeing where the graph crosses the x-axis.

```
> plot(sin(x)-sin(x+Pi/3),t=-2..2);
```



From this, we can see that the solutions in our range are $t = -\frac{5}{3}$, $t = -\frac{2}{3}$, $t = \frac{1}{3}$ and $t = \frac{4}{3}$. It is not too clear why these are the solutions, however.

```
> p1 := plot(sin(Pi*t),t=-2..2):
> r := evalf(sqrt(3)/2):
> p2 :=
   PLOT(
    CURVES(
     seq([[i+1/3,(-1)^i*r],[i+2/3,(-1)^i*r]],i=-2..1),
     COLOR(RGB,0,0,1),
     THICKNESS(3)
    )
   ):
> p3 :=
   PLOT(
    CURVES(
     seq([[i+1/3,0],[i+1/3,(-1)^i*r]],i=-2..1),
     COLOR(RGB,0,1,0),
     THICKNESS(3)
    )
   ):
> q := display(p1,p2,p3): q;
```



The picture below is more illuminating. If x is $\frac{\text{Pi}}{6}$ to the left of a peak of the sin graph, then $x + \frac{\text{Pi}}{3}$ is $\frac{\text{Pi}}{6}$ to the right of the same peak. The graph is symmetrical about the peaks, so we have $\sin(x) = \sin\left(x + \frac{\text{Pi}}{3}\right)$. The same thing works for troughs. The peaks and troughs are at the points $\left(n + \frac{1}{2}\right)\text{Pi}$, so our solutions are at $\left(n + \frac{1}{2}\right)\text{Pi} - \frac{\text{Pi}}{6}$, which is the same as $\left(n + \frac{1}{3}\right)\text{Pi}$.




```
> unassign('x','p1','p2','p3','q'):
```

Solve the equation $\cos(8\pi x)\sqrt{x-x^2} = \frac{1}{2}$ (with x between 0 and 1).

```
> y := cos(8*Pi*x)*sqrt(x-x^2);
```

$$y := \cos(8\,\pi\,x)\,\sqrt{-x^2+x} \tag{7.1}$$

From the picture, it is clear that the only solution is $x = \frac{1}{2}$.

```
> plot([y,1/2],x=0..1);
```



Maple is not able to work this out symbolically:

```
> solve(y=1/2,x);
```

$$\frac{RootOf\left(\cos(\_Z)\sqrt{8\pi\_Z-\_Z^2}-4\pi\right)}{8\pi} \qquad \textbf{(7.2)}$$

We can instead use the function **fsolve** to find a numerical approximation to the roots. This will usually work even if **solve** does not.

```
> fsolve(y=1/2,x);
```

$$0.5000000000 \qquad \textbf{(7.3)}$$

We can also use **fsolve** to solve the equation $y=\frac{1}{4}$, for which there is no easy exact solution.

```
> plot([y,1/4],x=0..1);
```




```
> fsolve(y=1/4,x);
```

0.2892634138 **(7.4)**

This gives only one of the six solutions. To find the rest, we must tell Maple roughly where to look. From the graph, we see that there is another solution close to $x = 0.8$. We can find the position more exactly as follows:

```
> fsolve(y=1/4,x=0.8);
```

0.7863971861 **(7.5)**

```
> unassign('y'):
```